Chapter 8:
Comparative dynamics in a model with a steady-state

Consider a very straight-forward dynamic model with an endogenous capital stock.

Sectors (Activities)

X_t	production of composite good in period t
I_t	production of new capital (investment) in period t
K_t	transforms capital into capital services and future capital

Commodities (Markets)

px (CXt)	price of X in period t
pr_t (CRt)	rental price of capital in period t
pk_t (CKt)	asset price of capital (price of a new capital good) in period t
pl_t (CLt)	price of labor in period t

Consumers

Infinitely lived representative consumer

δ	= rate of capital depreciation
ρ	= rate of time preference (discounting utility)
KE_t	= capital endowment at the beginning of a period
K_t	= capital stock for production at time t ($KE_t + I_t$)

Conditions for Steady-State Equilibrium:

(1) $$KE_{t+1} = KE_t \quad \Rightarrow \quad I_t = \delta KE_t$$

(2) rate of interest = ρ: $$pj_{t+1} = \frac{pj_t}{1+\rho} \quad \textit{for all goods } j = X, I, K$$

(3) $$pr_t = \left[1 - \frac{1-\delta}{1+\rho}\right] pk_t$$ relationship between asset and rental prices

(4) $$KE_{t+1} = (1-\delta)K_t = (1-\delta)(KE_t + I_t)$$ (KE: capital endowment)

$$pk_{t+1}KE_{t+1} = (1-\delta)\frac{pk_t}{1+\rho}K_t = (1-\delta)\frac{(pk_tKE_t + pk_tI_t)}{1+\rho}$$

	Xt	It	Kt	CONS	
CXt	200			-200	0
CRt	-100		100[3]		0
CKt		40	-400[2]	360[1]	0
CLt	-100	-40		140	0
CKt+1			300[4]	-300	0

Parameters: RHO = 0.2, DELTA = 0.1

Prices: CX0=CR0=CL0=1: CK0=4 CK1=CK0/(1+RHO)= 3.3333
CR0 = (1 - (1-DELTA)/(1+RHO))*CK0 = (1/4)*CK0

1 360 = 90 units at CK0 = 4
2 400 = 100 units at CK0 = 4
3 100 = 100 units at rental price = 1
4 300 = undepreciated capital (1-delta)*100 = 90
at a price of CK1 = 1/(1+RHO) = 4/1.2 = 3.3333
300 = (1-DELTA)*4*100/(1+RHO)= 300

The amount 360 - 300 = 60 can be thought of as net rental income: rental income (90) minus the cost of replacing depreciated capital: 9*CK0/(1+RHO) = 30.

Problem: Suppose we want to represent this infinite-horizon problem as a finite dimension complementarity problem

Approaching the last period the consumer would have no incentive to accumulate capital and would want to run down the capital stock.

(1) Assume a finite number of periods plus a terminal period.

(2) Assume an extra dummy agent (God? But don’t want to offend anyone)

(3) Assume that the dummy agent is endowed with an extra good “Heaven”

(4) Assume that the dummy agent will only sell Heaven in exchange for terminal period capital (does not demand any other good)

(5) Assume that the representative agent has a demand for heaven

(6) Use a tax/subsidy on heaven to ensure that the asset/rental price relationship holds on terminal capital (so that the economy is forced onto the steady-state path at terminal time)

Terminal period

	Xt	It	Kt	CONS	DUMMY	
CXt	200			-200	0	0
CRt	-100		100		0	0
CKt		40	-400	360	0	0
CLt	-100	-40		140	0	0
CKt+1			300		-300	0
Heaven				-300	300	0
	0	0	0	0		

```
SETS T  /1*25/;

PARAMETERS
 DELTA
 RHO
 PV
 TERM
 RTERM
 INITK
 R(T)
 D(T)
 PVUTIL
 TLAST(T)
 TFIRST(T)
 SOLUTION(T,*)
 CONSUME(T)
 INVEST(T)
 KSTOCK(T);


RHO = 0.2;
DELTA = 0.1;
INITK = 90;
```

```
TERM = CARD(T);
RTERM = (1/(1+RHO))**(CARD(T) - 1);;
R(T) = (1/(1+RHO))**(ORD(T)-1);
D(T) = (1-DELTA)**(ORD(T) - 1);
PV = 200*SUM(T, R(T)) + 90*(4*RTERM/(1+RHO));
TLAST(T) = 0;
TLAST('25') = 1;
TFIRST('1') = 1;

$ONTEXT
$MODEL: BASIC

$SECTORS:
 X(T)
 I(T)
 K(T)
 U

$COMMODITIES:
 CX(T)
 CR(T)
 CK(T)
 CL(T)
```

```
 CKT
 CU
 HEAVEN

$CONSUMERS:
 CONS
 DUMMY

$AUXILIARY:
 TRANS

$PROD:K(T)
 O:CK(T+1)          Q:(100*(1-DELTA))    P:(4*R(T+1))
 O:CKT$TLAST(T)     Q:(100*(1-DELTA))    P:(4*R(T)/(1+RHO))
 O:CR(T)            Q:100                P:(R(T))
 I:CK(T)            Q:100                P:(4*R(T))

$PROD:I(T)
 O:CK(T)         Q:10
 I:CL(T)         Q:40
```

```
$PROD:X(T) s:1
 O:CX(T)         Q:200
 I:CL(T)         Q:100
 I:CR(T)         Q:100

$PROD:U  s:1 a:2
 O:CU   Q:PV
 I:CX(T)   Q:200  P:R(T)  a:
 I:HEAVEN  Q:90   P:(4*RTERM/(1+RHO))   A:CONS  N:TRANS

$DEMAND:CONS
 D:CU                      Q:PV
 E:CL(T)                   Q:140
 E:CK(T)$TFIRST(T)         Q:INITK

$DEMAND:DUMMY
 D:CKT           Q:90
 E:HEAVEN        Q:90

$CONSTRAINT: TRANS
  CR('25') =E= (1 - (1-DELTA)/(1+RHO))*CL('25')*4;
```

```
$OFFTEXT
$SYSINCLUDE MPSGESET BASIC

TRANS.UP = +INF;
TRANS.LO = -INF;

CX.L(T) = R(T);
CL.L(T) = R(T);
CR.L(T) = R(T);
CK.L(T) = 4*R(T);
CKT.L = 4*R('25')/(1+RHO);
HEAVEN.L = 4*R('25')/(1+RHO);
TRANS.L = 0;

*BASIC.ITERLIM = 0;

$INCLUDE BASIC.GEN
SOLVE BASIC USING MCP;

PVUTIL = SUM(T, X.L(T)*R(T)) + (X.L('25')*R('25'))/RHO;

DISPLAY PVUTIL;
```

```
CONSUME(T) = X.L(T);
INVEST(T) = I.L(T);
KSTOCK(T) = K.L(T);
SOLUTION(T,"X") = X.L(T);
SOLUTION(T,"I") = I.L(T);
SOLUTION(T,"K") = K.L(T);

$LIBINCLUDE XLDUMP SOLUTION SOL2.xls SHEET1!A2



INITK = 30;

$INCLUDE BASIC.GEN
SOLVE BASIC USING MCP;

PVUTIL = SUM(T, X.L(T)*R(T)) + (X.L('25')*R('25'))/RHO;

DISPLAY PVUTIL;


CONSUME(T) = X.L(T);
INVEST(T) = I.L(T);
KSTOCK(T) = K.L(T);
```

```
SOLUTION(T,"X") = X.L(T);
SOLUTION(T,"I") = I.L(T);
SOLUTION(T,"K") = K.L(T);

$LIBINCLUDE XLDUMP SOLUTION SOL2.xls SHEET1!F2

* make people more patient, raise rho to 0.1

INITK = 90;
RHO = 0.1;

RTERM = (1/(1+RHO))**(CARD(T) - 1);;
R(T) = (1/(1+RHO))**(ORD(T)-1);
D(T) = (1-DELTA)**(ORD(T) - 1);
PV = 200*SUM(T, R(T)) + 90*(4*RTERM/(1+RHO))
;
$INCLUDE BASIC.GEN
SOLVE BASIC USING MCP;

PVUTIL = SUM(T, X.L(T)*R(T)) + (X.L('25')*R('25'))/RHO;

DISPLAY PVUTIL;
```

```
CONSUME(T) = X.L(T);
INVEST(T) = I.L(T);
KSTOCK(T) = K.L(T);
SOLUTION(T,"X") = X.L(T);
SOLUTION(T,"I") = I.L(T);
SOLUTION(T,"K") = K.L(T);

$LIBINCLUDE XLDUMP SOLUTION SOL2.xls SHEET1!K2
```